How to make free resolutions with Macaulay2

Chris Peterson and Hirotachi Abo

1. What are syzygies?

Let k be a field, let $R = k[x_0, \ldots, x_n]$ be the homogeneous coordinate ring of $\mathbb{P}^n$ and let X be a projective variety in $\mathbb{P}^n$. Consider the ideal $I(X)$ of X. Assume that $\{f_0, \ldots, f_t\}$ is a generating set of $I(X)$ and that each polynomial f_i has degree d_i. We can express this by saying that we have a surjective homogenous map of graded S-modules:

$$\bigoplus_{i=0}^{t} R(-d_i) \to I(X),$$

where $R(-d_i)$ is a graded R-module with grading shifted by $-d_i$, that is,

$$R(-d_i)_k = R_{k-d_i}.$$

In other words, we have an exact sequence of graded R-modules:



where $\mathbf{F} = (f_0, \ldots, f_t)$.

Example 1. Let $R = k[x_0, x_1, x_2]$. Consider the union P of three points $[0 : 0 : 1]$, $[1 : 0 : 0]$ and $[0 : 1 : 0]$. The corresponding ideals are (x_0, x_1), (x_1, x_2) and (x_2, x_0). The intersection of these ideals are (x_1x_2, x_0x_2, x_0x_1). Let $I(P)$ be the ideal of P. In `Macaulay2`, the generating set $\{x_1x_2, x_0x_2, x_0x_1\}$ for $I(P)$ is described as a 1×3 matrix with `gens`:

```
i1 : KK=QQ

o1 = QQ

o1 : Ring
```

```
  --  the class of all rational numbers

i2 : ringP2=KK[x_0,x_1,x_2]

o2 = ringP2

o2 : PolynomialRing

i3 : P1=ideal(x_0,x_1); P2=ideal(x_1,x_2); P3=ideal(x_2,x_0);

o3 : Ideal of ringP2

o4 : Ideal of ringP2

o5 : Ideal of ringP2

i6 : P=intersect(P1,P2,P3)

o6 = ideal (x x , x x , x x )
             1 2   0 2   0 1

o6 : Ideal of ringP2

i7 : gens P

o7 = | x_1x_2 x_0x_2 x_0x_1 |

                     1            3
o7 : Matrix ringP2  <--- ringP2
```

Definition. The kernel M_1 of $\mathbf{F}$ will be a graded R-module generated by the *relations* among $f_0, \dots, f_{t-1}$ and f_t. To be more precise, M_1 is a module of $(t+1)$-tuples $\mathbf{G} = (g_0, \dots, g_t)$ such that $\mathbf{F} \cdot \mathbf{G}^T = 0$. Note that M_1 is finitely generated, unless $t = 0$ (why?). An element of M_1 is called a *syzygy* of $I(P)$), and M_1 is called the *module of syzygies* of $I(P)$. Let $\{\mathbf{G}_0, \cdots, \mathbf{G}_r\}$ be a set of generators for M_1, where $\mathbf{G}_j = (g_0^j, \dots, g_t^j)$. Then we have the following figure:

$$\begin{array}{ccccccccc}
\bigoplus_{j=0}^{r} R(-d'_j) & & \xrightarrow{(\mathbf{G}_0,\dots,\mathbf{G}_r)} & & \bigoplus_{i=0}^{t} R(-d_i) & \longrightarrow & R & \longrightarrow \Gamma(X) \longrightarrow & 0, \\
& \searrow & & \nearrow & & & & & \\
& & M_1 & & & & & & \\
& \nearrow & & \searrow & & & & & \\
0 & & & & 0 & & & &
\end{array}$$

where $\deg(g_k^j) + d_k = d'_j$ for all $k = 0, \dots, t$.

Example 2. Let P be the union of three points given in Example 1, and let $I(P)$ be its ideal in R. Let's compute the syzygy module of $I(P)$. The command `syz` computes the syzygy module M_1 for a given ideal, or more generally a given module M. The input is the matrix whose columns generates M. The command `syz` returns the matrix whose columns generates M_1:

```
 i8 : syz gens P

o8 = {2} | x_0  0    |
     {2} | 0    x_1  |
     {2} | -x_2 -x_2 |

                     3            2
o8 : Matrix ringP2  <--- ringP2
```

The kernel of $(\mathbf{G}_0, \dots, \mathbf{G}_r)$ is again a finitely generated R-module. Consider the relations among $\mathbf{G}_0, \dots, \mathbf{G}_r$. They generate a new finitely generated graded R-module M_2.

Example 3. We have shown that for $I(P)$, the corresponding module M_1 is represented by the following matrix:

$$\begin{array}{ccccccc}
\cdots \longrightarrow & 2R(-3) & & \xrightarrow{\begin{pmatrix} x_0 & 0 \\ 0 & x_1 \\ -x_2 & -x_2 \end{pmatrix}} & & 3R(-2) & \longrightarrow \cdots, \\
& & \searrow & & \nearrow & & \\
& & & M_1 & & & \\
& & \nearrow & & \searrow & & \\
& 0 & & & & 0 &
\end{array}$$

where

$$mR(-d) = \overbrace{R(-d) \oplus \cdots \oplus R(-d)}^{m}.$$

Let $\mathbf{G} = (\mathbf{G}_0, \mathbf{G}_1)$ be the matrix whose columns generate M_1. What is the kernel of $\mathbf{G}$? Let f and g be homogeneous polynomials in R. Assume that the R-linear combination $f \cdot \mathbf{G}_0 + g \cdot \mathbf{G}_1$ is zero. This relation gives rise to the equations $f \cdot z = 0$ and $g \cdot y = 0$. So $f = g = 0$, because both the polynomials f and g are nonzero-divisors in R. Thus we get the exact sequence of type:

$$0 \to 2R(-3) \to 3R(-2) \to R \to \Gamma(P) \to 0.$$

Indeed,

```
o8 : Matrix ringP2  <--- ringP2

i9 : syz (syz gens P)

o9 = 0

                     2
o9 : Matrix ringP2  <--- 0
```

Note that this sequence is exact at the level of vector spaces:

$$0 \to 2R(-3)_d \to 3R(-2)_d \to R_d \to \Gamma(P)_d \to 0.$$

This exact sequence can be used to compute the Hilbert function $\mathrm{HF}(I(P), d)$:

$$\begin{aligned}
\dim_k(\Gamma(V))_d &= \dim_k(R_d) - 3 \cdot \dim_k(R(-2)_d) + 2 \cdot \dim_k(R(-3)) \\
&= \binom{d+2}{2} - 3 \cdot \binom{d}{2} + 2 \cdot \binom{d-1}{2} = 3.
\end{aligned}$$

2. Free resolutions.

The operation we performed can be repeated over and over again, and we can make a long exact sequence of free R-modules:

$$\cdots \to F_m \to \cdots \to F_2 \to F_1 \to R \to \Gamma(V) \to 0.$$

This sequence is called a *free resolution* of $\Gamma(V)$. Obviously, a free resolution of a graded R-module M can be also computed in the same manner:

$$\cdots \to F_m \to \cdots \to F_2 \to F_1 \to F_0 \to M \to 0.$$

We say that such a resolution has *length* m if $F_m \neq 0$ but $F_i = 0$ for all $i > m$.

A natural question is: "Can we make a finite free resolution for every finitely generated graded R-module?" If the answer is yes, we can compute the Hilbert function as follows:

$$\mathrm{HF}(M, d) = \sum_{i=0}^{m} (-1)^i \mathrm{HF}(F_i, d).$$

We cannot expect the answer to be yes over a non-polynomial ring. But Hilbert showed the following theorem:

Theorem (Hilbert Syzygies Theorem, 1890). Every finitely generated graded M R-module has a finite free resolution:

$$0 \to F_m \to \cdots \to F_2 \to F_1 \to F_0 \to M \to 0.$$

Moreover, we may take $m \leq n + 1$.

This theorem motivates us to compute a free resolution of a finitely generated module over a polynomial ring.

Example 4. Let's compute the free resolution of $\Gamma(P)$! In `Macaulay2`, we can compute a free resolution of a given module with the command `resolution` or the synonym `res`. These commands take as an input an ideal or a module and returns a free resolution.

Remark.
(1) The module itself is not displayed in the free resolution.
(2) In case the input is an ideal I, `res` computes the free resolution of R/I, but not I.

```
i22 : fP=res P

             1            3            2
o22 = ringP2  <-- ringP2  <-- ringP2  <-- 0

      0            1            2            3

o22 : ChainComplex
```

We can see all the differentials by adding the suffix `.dd` to the name of a free resolution:

```
i23 : fP.dd

                1                                  3
o23 = 0 : ringP2  <--------------------------- ringP2  : 1
                    | x_0x_1 x_0x_2 x_1x_2 |

                3                          2
      1 : ringP2  <-------------------- ringP2  : 2
                    {2} | -x_2 0    |
                    {2} | x_1  -x_1 |
                    {2} | 0    x_0  |

                2
      2 : ringP2  <----- 0 : 3
                    0

o23 : ChainComplexMap
```

The free resolution given above is of the following form:

$$0 \to 2R(-3) \to 3R(-2) \to R \to \Gamma(V) \to 0.$$

The command `betti` gives information about the resolution looks like:

```
i12 : betti fP

o12 = total: 1 3 2
          0: 1 . .
          1: . 3 2
```